
Abstract. Functional reactive programming, or FRP, is a paradigm for
programming hybrid systems { i.e., systemscontaining a combination of
both contin uous and discrete components { in a high-level, declarative
way. The key ideas in FRP are its notions of contin uous, time-varying
values, and time-ordered sequencesof discrete events.
Yampa is an instantiation of FRP as a domain-speci�c language embed-
ded in Haskell. This paper describes Yampa in detail, and shows how
it can be used to program a particular kind of hybrid system: a mobile
robot. Becauseperformance is critical in robotic programming, Yampa
usesarrows (a generalization of monads) to create a disciplined style of
programming with time-varying values that helps ensure that common
kinds of time- and space-leaksdo not occur.
No previous experiencewith robots is expected of the reader, although a
basic understanding of physics and calculus is assumed.No knowledge of
arrows is required either, although we assumea good working knowledge
of Haskell.

This paper is dedicated in memory of
Edsger W. Dijkstra

for his in
uential insight that mathematical logic is and
must be the basis for sensiblecomputer program construction.

1 In tro duction

Can functional languagesbe used in the real world, and in particular for real-
time systems?More speci�cally , can the expressivenessof functional languages
beusedadvantageouslyin such systems,and canperformanceissuesbeovercome
at least for the most common applications?

For the past several yearswe have been trying to answer thesequestionsin
the a�rmativ e. We have developed a generalparadigm called functional reactive

programming that is well suited to programming hybrid systems, i.e. systems
with both continuous and discrete components. An excellent example of a hy-
brid system is a mobile robot. From a physical perspective, mobile robots have
continuous components such as voltage-controlled motors, batteries, and range
�nders, aswell asdiscretecomponents such asmicroprocessors,bumper switches,
and digital communication. More importantly , from a logical perspective, mobile
robots have continuous notions such as wheel speed, orientation, and distance
from a wall, as well as discrete notions such as running into another object,
receiving a message,or achieving a goal.

Functional reactive programming was �rst manifested in Fran, a domain
speci�c language(DSL) for graphics and animation developed by Conal Elliott
at Microsoft Research [5,4]. FRP [13,8,16] is a DSL developed at Yale that is
the \essence" of Fran in that it exposesthe key conceptswithout bias toward
application speci�cs. FAL [6], Frob [11,12], Fvision [14], and Fruit [2] are four
other DSLs that we have developed, each embracing the paradigm in ways suited
to a particular application domain. In addition, we have pushed FRP toward
real-time embeddedsystemsthrough several variants including Real-Time FRP
and Event-Driven FRP [18,17,15].

The core ideas of functional reactive programming have evolved (often in
subtle ways) through thesemany languagedesigns,culminating in what we now
call Yampa, which is the main topic of this paper.1 Yampa is a DSL embedded
in Haskell and is a re�nement of FRP. Its most distinguishing feature is that the
coreFRP conceptsare represented using arrows [7], a generalizationof monads.
The programming discipline induced by arrows prevents certain kinds of time-
and space-leaksthat are commonin genericFRP programs, thus making Yampa
more suitable for systemshaving real-time constraints.

Yampa has beenusedto program real industrial-strength mobile robots [10,
8]2, building on earlier experience with FRP and Frob [11,12]. In this paper,
however, we will usea robot simulator. In this way, the readerwill be able to run
all of our programs, aswell as new onesthat shemight write, without having to
buy a $10,000robot. All of the code in this paper, and the simulator itself, are
available via the Yampa home pageat www.haskell.org/yampa .

The simulated robot, which we refer to as a simbot, is a di�er ential drive
robot, meaning that it has two wheels, much like a cart, each driven by an
independent motor. The relative velocity of these two wheelsthus governs the
turning rate of the simbot; if the velocities are identical, the simbot will go
straight. The physical simulation of the simbot includes translational inertia,
but (for simplicit y) not rotational inertia.

The motors are what makes the simbot go; but it also has several kinds of
sensors.First, it has a bumper switch to detect when the simbot gets \stuck."

1 Yampa is a riv er in Colorado whoselong placid sectionsare occasionally interrupted
by turbulen t rapids, and is thus a good metaphor for the contin uous and discrete
components of hybrid systems. But if you prefer acronyms, Yampa was started at
YAle, ended in Arro ws, and had Much Programming in between.

2 In these two earlier papers we referred to Yampa as AFRP .



That is, if the simbot runs into something, it will just stop and signal the pro-
gram. Second,it hasa range �nder that can determine the nearestobject in any
given direction. In our exampleswe will assumethat the simbot hasindependent
range �nders that only look forward, backward, left, and right, and thus we will
only query the range �nder at thesefour angles.Third, the simbot has what we
call an \animate object tracker" that gives the location of all other simbots, as
well as possibly somefree-moving balls, that are within a certain distance from
the simbot. You can think of this tracker as modelling either a visual subsystem
that can seethese objects, or a communication subsystemthrough which the
simbots and balls shareeach others' coordinates. Each simbot also has a unique
ID and a few other capabilities that we will intro duce as we needthem.

2 Yampa Basics

The most important conceptunderlying functional reactive programming is that
of a signal: a continous, time-varying value. One can think of a signal as having
polymorphic type:

Signal a = Time -> a

That is, a value of type Signal a is a function mapping suitable valuesof time
(Double is usedin our implementation) to a value of type a. Conceptually, then,
a signal s's value at sometime t is just s(t).

For example, the velocity of a di�eren tial drive robot is a pair of numbers
representing the speedsof the left and right wheels. If the speedsare in turn
represented as type Speed, then the robot's velocity can be represented as type
Signal (Speed,Speed). A program controlling the robot must thereforeprovide
such a value as output.

Being able to de�ne and manipulate continuous values in a programming
languageprovides great expressive power. For example, the equationsgoverning
the motion of a di�eren tial drive robot [3] are:

x(t) = 1
2

Rt
0 (vr (t) + vl (t)) cos(� (t)) dt

y(t) = 1
2

Rt
0 (vr (t) + vl (t)) sin(� (t)) dt

� (t) = 1
l

Rt
0 (vr (t) � vl (t)) dt

where x(t), y(t), and � (t) are the robot's x and y coordinates and orientation,
respectively; vr (t) and vl (t) are the right and left wheelspeeds,respectively; and
l is the distancebetweenthe two wheels.In FRP theseequationscan be written
as:

x = (1/2) * integral ((vr + vl) * cos theta)
y = (1/2) * integral ((vr + vl) * sin theta)
theta = (1/l) * integral (vr - vl)



All of the values in this FRP program are implicitly time-varying, and thus
the explicit time t is not present. 3 Nevertheless,the direct correspondencebe-
tween the physical equations (i.e. the speci�cation) and the FRP code (i.e. the
implementation) is very strong.

2.1 Arro wized FRP

Although quite general, the concept of a signal can lead to programs that have
conspicuoustime- and space-leaks,4 for reasonsthat are beyond the scope of
this paper. Earlier versions of Fran, FAL, and FRP used various methods to
make this performance problem lessof an issue,but ultimately they all either
su�ered from the problem in one way or another, or intro duced other problems
as a result of �xing it.

In Yampa, the problem is solved in a more radical way: signalsare simply not
allowed as �rst-class values! Instead, the programmer has accessonly to signal
transformers, or what we prefer to call signal functions. A signal function is just
a function that maps signals to signals:

SF a b = Signal a -> Signal b

However, the actual representation of the type SF in Yampa is hidden (i.e. SF
is abstract), so one cannot directly build signal functions or apply them to sig-
nals. Instead of allowing the userto de�ne arbitrary signal functions from scratch
(which makesit all too easyto intro ducetime- and space-leaks),we provide a set
of primitiv e signal functions and a set of special composition operators (or \com-
binators") with which more complex signal functions may be de�ned. Together,
these primitiv e valuesand combinators provide a disciplined way to de�ne sig-
nal functions that, fortuitously , avoids time- and space-leaks.We achieve this by
structuring Yampa basedon arrows, a generalizationof monadsproposedin [7].
Speci�cally , the type SFis made an instance of the Arrow class.

Sobroadly speaking, a Yampa program expressesthe composition of a possi-
bly large number of signal functions into a composite signal function that is then
\run" at the top level by a suitable interpreter. A good analogy for this idea is a
state or IO monad, where the state is hidden, and a program consistsof a linear
sequencingof actions that are eventually run by an interpreter or the operating
system.But in fact arrows are more generalthan monads,and in particular the
composition of signal functions does not have to be completely linear, as will
be illustrated shortly. Indeed, becausesignal functions are abstract, we should

3 This implies that the sine, cosine,and arithmetic operators are over-loaded to handle
signals properly.

4 A time-leak in a real-time system occurs whenever a time-dependent computation
falls behind the current time becauseits value or e�ect is not neededyet, but then
requires \catc hing up" at a later point in time. This catching up processcan take
an arbitrarily long time, and may or may not consumespaceas well. It can destroy
any hope for real-time behavior if not managed properly.



be concernedthat we have a su�cien t set of combinators to composeour signal
functions without lossof expressive power.

We will motivate the set of combinators usedto composesignal functions by
using an analogy to so-called\p oint-free" functional programming (an example
of which is the Bird/Meertens formalism [1]). For the simplest possibleexample,
supposethat f1 :: A -> B and f2 :: B -> C. Then instead of writing:

g :: A -> C
g x = f2 (f1 x)

we can write in a point-free style using the familiar function composition oper-
ator:

g = f2 . f1

This code is \p oint-free" in that the values(points) passedto and returned from
a function are never directly manipulated.

To do this at the level of signal functions, all we needis a primitiv e operator
to \lift" ordinary functions to the level of signal functions:

arr :: (a -> b) -> SF a b

and a primitiv e combinator to composesignal functions:

(>>>) :: SF a b -> SF b c -> SF a c

We can then write:

g' :: SF A C
g' = arr g

= arr f1 >>> arr f2

Note that (>>>) actually represents reversefunction composition, and thus its
arguments are reversedin comparisonto (.) .

Unfortunately , most programs are not simply linear compositions of func-
tions, and it is often the casethat more than oneinput and/or output is needed.
For example,supposethat f1 :: A -> B, f2 :: A -> Cand we wish to de�ne
the following in point-free style:

h :: A -> (B,C)
h x = (f1 x, f2 x)

Perhaps the simplest way is to de�ne a combinator:

(&) :: (a->b) -> (a->c) -> a -> (b,c)
(f1 & f2) x = (f1 x, f2 x)

which allows us to de�ne h simply as:

h = f1 & f2



In Yampa there is a combinator (&&&) :: SF a b -> SF a c -> SF a (b,c)
that is analogousto &, thus allowing us to write:

h' :: SF A (B,C)
h' = arr h

= arr f1 &&&arr f2

As another example, suppose that f1 :: A -> B and f2 :: C -> D. One
can easily write a point-free version of:

i :: (A,C) -> (B,D)
i (x,y) = (f1 x, f2 y)

by using (&) de�ned above and Haskell's standard fst and snd operators:

i = (f1 . fst) & (f2 . snd)

For signal functions, all we need are analogousversionsof fst and snd, which
we can achieve via lifting:

i' :: SF (A,C) (B,D)
i' = arr i

= arr (f1 . fst) &&&arr (f2 . snd)
= (arr fst >>> arr f1) &&&(arr snd >>> arr f2)

The \argument wiring" pattern captured by i' is in fact a commonone,and
thus Yampa provides the following combinator:

(***) :: SF b c -> SF b' c' -> SF (b,b') (c,c')
f *** g = (arr fst >>> f) &&&(arr snd >>> g)

so that i' can be written simply as:

i' = arr f1 *** arr f2

g' , h' , and i' were derived from g, h, and i , respectively, by appealing to
one's intuition about functions and their composition. In the next section we
will formalize this using type classes.

2.2 The Arro w Class

Onecould goon and on in this manner,adding combinators asthey areneededto
solveparticular \argument wiring" problems,but it behoovesusat somepoint to
ask if there is a minimal universal set of combinators that is su�cien t to express
all possible wirings. Note that so far we have intro duced three combinators {
arr , (>>>) , and (&&&) { without de�nitions, and a fourth { (***) { wasde�ned
in terms of these three. Indeed these three combinators constitute a minimal
universal set.

However, this is not the only minimal set. In fact, in de�ning the original
Arrow class,Hughesinstead chosethe set arr , (>>>) , and first :



class Arrow a where
arr :: (b -> c) -> a b c
(>>>) :: a b c -> a c d -> a b d
first :: a b c -> a (b,d) (c,d)

where first is analogousto the following function de�ned at the ordinary func-
tion level:

firstfun f = \(x,y) -> (f x, y)

In Yampa, the type SF is an instance of classArrow, and thus these types are
consistent with what we presented earlier. To help seehow this set is miminal,
here are de�nitions of second and (&&&) in terms of the Arrow classmethods:

second :: Arrow a => a b c -> a (d,b) (d,c)
second f = arr swap >>> first f >>> arr swap

where swap pr = (snd pr, fst pr)

(&&&) :: Arrow a => a b c -> a b d -> a b (c,d)
f &&&g = arr (\b -> (b,b)) >>> (first f >>> second g)

In addition, here is an instance declaration that shows how Haskell's normal
function type can be treated as an arrow:

instance Arrow (->) where
arr f = f
f >>> g = g . f
first f = \(b,d) -> (f b, d)

With this instance declaration, the derivations of g' , h' , and i' in the previous
section can be formally justi�ed.

Exercise 1. De�ne (a) first in terms of just arr , (>>>) , and (&&&), (b) (***)
in terms of just first , second, and (>>>) , and (c) (&&&) in terms of just arr ,
(>>>) , and (***) .

2.3 Commonly Used Com binators

In practice, it is better to think in terms of a commonly-usedset of combinators
rather than a minimal set. Figure 1 shows a set of eight combinators that we use
often in Yampa programming, along with the graphical \wiring of arguments"
that �v e of them imply.

Yampa also provides many convenient library functions that facilitate pro-
gramming in the arrow framework. Here are somethat we will use later in this
paper:

identity :: SF a a
constant :: b -> SF a b
time :: SF a Time



arr :: Arrow a => (b -> c) -> a b c
(>>>) :: Arrow a => a b c -> a c d -> a b d
(<<<) :: Arrow a => a c d -> a b c -> a b d
first :: Arrow a => a b c -> a (b,d) (c,d)
second :: Arrow a => a b c -> a (d,b) (d,c)
(***) :: Arrow a => a b c -> a b' c' -> a (b,b') (c,c')
(&&&) :: Arrow a => a b c -> a b c' -> a b (c,c')
loop :: Arrow a => a (b,d) (c,d) -> a b c

f

(a) arr f

�����

�����

(b) sf1 >>> sf2

���




(c) first sf

	�
��



	�
��




(d) sf1 &&&sf2


��




(e) loop sf

Fig. 1. Commonly Used Arro w Combinators



The identity signal function is analogousto the identit y function in Haskell,
and in fact is equivalent to arr id . The constant function is useful for generat-
ing constant signal functions, is analogousto Haskell's const function, and in fact
is equivalent to arr . const . Finally, time is a signal function that yields the
current time, and is equivalent to constant 1.0 >>> integral , whereintegral
is a pre-de�ned Yampa signal function with type:5

integral :: SF Double Double

Yampa also de�nes a derivative signal function.
It is important to note that somesignal functions are stateful, in that they

accumulate information over time. integral is a perfect exampleof such a func-
tion: by de�nition, it sums instantaneuousvaluesof a signal over time. Stateful
signal functions cannot be de�ned using arr , which only lifts pure functions to
the level of arrows. Stateful functions must either be pre-de�ned or be de�ned
in terms of other stateful signal functions.

Stated another way, stateful signal functions such as integration and di�er-
entiation depend intimately on the underlying time-varying semantics, and so
do not have analogousunlifted forms. Indeed, it is soeasyto lift unary functions
to the level of signal functions that there is generally no need to provide spe-
cial signal function versionsof them. For example, instead of de�ning a special
sinSF, cosSF, etc., we can just use arr sin , arr cos, etc. Furthermore, with
the binary lifting operator:

arr2 :: (a->b->c) -> SF (a,b) c
arr2 = arr . uncurry

we can also lift binary operators. For example,arr2 (+) has type
Numa => SF (a,a) a.

2.4 A Simple Example

To seeall of this in action, consider the FRP code presented earlier for the
coordinates and orientation of the mobile robot. We will rewrite the code for the
x-coordinate in Yampa (leaving the y-coordinate and orientation asan exercise).

Supposethere are signal functions vrSF, vlSF :: SF SimbotInput Speed
and thetaSF :: SF SimbotInput Angle. The typeSimbotInput represents the
input state of the simbot, which we will have much more to say about later. With
thesesignal functions in hand, the previous FRP code for x:

x = (1/2) * integral ((vr + vl) * cos theta)

can be rewritten in Yampa as:

5 This function is actually overloaded for any vector space,but that doesnot concern
us here, and thus we have specialized it to Double.



xSF :: SF SimbotInput Distance
xSF = let v = (vrSF &&&vlSF) >>> arr2 (+)

t = thetaSF >>> arr cos
in (v &&&t) >>> arr2 (*) >>> integral >>> arr (/2)

Exercise 2. De�ne signal functions ySFand thetaSF in Yampa that correspond
to the de�nitions of y and theta , respectively, in FRP.

2.5 Arro w Syn tax

Although we have achieved the goal of preventing direct accessto signals, one
might argue that we have lost the clarit y of the original FRP code: the code
for xSF is certainly more di�cult to understand than that for x. Most of the
complexity is due to the needto wire signal functions together using the various
pairing/unpairing combinators such as (&&&) and (***) . Precisely to address
this problem, Paterson [9] hassuggestedthe useof special syntax to make arrow
programming more readable, and has written a preprocessorthat converts the
syntactic sugar into conventional Haskell code. Using this special arrow syntax,
the above Yampa code for xSFcan be rewritten as:

xSF' :: SF SimbotInput Distance
xSF' = proc inp -> do

vr <- vrSF -< inp
vl <- vlSF -< inp
theta <- thetaSF -< inp
i <- integral -< (vr+vl) * cos theta
returnA -< (i/2)

Although not quite as readableas the original FRP de�nition of x, this code is
far better than the unsugaredversion. There are several things to note about
the structure of this code:

1. The syntax proc pat -> ... is analogousto a Haskell lambda expression
of the form \ pat -> ... , except that it de�nes a signal function rather
than a normal Haskell function.

2. In the syntax pat <- SFexpr -< expr, the expressionSFexpr must be a
signal function, say of type SF T1 T2, in which caseexpr must have type
T1 and pat must have type T2. This is analogousto pat = expr1 expr2 in a
Haskell let or where clause,in which caseif expr1 has type T1 -> T2, then
expr2 must have type T1 and pat must have type T2.

3. The overall syntax:

proc pat -> do
pat1 <- SFexpr1 -< expr1
pat2 <- SFexpr2 -< expr2
� � �
returnA -< expr



de�nes a signal function. If pat has type T1 and expr has type T2, then the
type of the signal function is SF T1 T2. In addition, any variable bound by
one of the patterns pati can only be used in the expressionexpr or in an
expressionexprj where j > i . In particular, it cannot be used in any of the
signal function expressionsSFexpri .

It is important to note that the arrow syntax allows one to get a handle
on a signal's values (or samples), but not on the signals themselves. In other
words, �rst recalling that a signal function SF a b can be thought of as a type
Signal a -> Signal b, which in turn can be thought of as type
(Time -> a) -> (Time -> b) , the syntax allows getting a handle on valuesof
type a and b, but not on valuesof type Time -> a or Time -> b.

Figure 2(a) is a signal 
ow diagram that precisely represents the wiring im-
plied by the sugaredde�nition of xSF' . (It alsore
ects well the data dependencies
in the original FRP program for x.) Figure 2(b) shows the samediagram, except
that it hasbeenoverlaid with the combinator applications implied by the unsug-
ared de�nition of xSF (for clarit y, the lifting via arr of the primitiv e functions
{ i.e. those drawn with circles { is omitted). Thesediagrams demonstrate nicely
the relationship betweenthe sugaredand unsugaredforms of Yampa programs.

Exercise 3. Rewrite the de�nitions of ySFand thetaSF from the previous exer-
ciseusing the arrow syntax. Also draw their signal 
o w diagrams.

2.6 Discrete Ev ents and Switc hing

Most programming languageshave somekind of conditional choice capability,
and Yampa is no exception. Indeed, given signal functions flag :: SF a Bool
and sfx, sfy :: SF a b, then the signal function:

sf :: SF a b
sf = proc i -> do

x <- sfx -< i
y <- sfy -< i
b <- flag -< i
returnA -< if b then x else y

behaves like sfx whenever flag yields a true value, and like sfy whenever it
yields false.

However, this is not completely satisfactory, becausethere are many situ-
ations where one would prefer that a signal function switch into, or literally
become,someother signal function, rather than continually alternate between
two signal functions basedon the value of a boolean. Indeed, there is often a
successionof new signal functions to switch into as a successionof particular
events occurs,much like state changesin a �nite state automaton. Furthermore,
we would like for thesenewly invoked signal functions to start afresh from time
zero, rather than being signal functions that have been \running" since the

program began. This relates precisely to the issue of \statefulness" that was
previously discussed.

This advancedfunctionalit y is achieved in Yampa using eventsand switching
combinators.

In previous versions of FRP, including Fran, Frob, and FAL, a signi�cant
distinction was made betweencontinuous valuesand discrete events. In Yampa
this distinction is not asgreat. Events in Yampa are just abstract valuesthat are
isomorphic to Haskell's Maybedata type. A signal of type Signal (Event b)
is called an event stream, and is a signal that, at any point in time, yields
either nothing or an event carrying a value of type b. A signal function of type
SF a (Event b) generatesan event stream, and is called an event source.

Note: Although event streamsand continuousvaluesare both represented as
signals in Yampa, there are important semantic di�erences between them. For
example, improper useof events may lead to programs that are not convergent,
or that allow the underlying sampling rate to \show through" in the program's
behavior. Semantically speaking, event streams in Yampa should not be \in-
�nitely dense" in time; practically speaking, their frequency should not exceed
the internal sampling rate unlessbu�ering is provided.6

As an exampleof a well-de�ned event source,the signal function:

rsStuck :: SF SimbotInput (Event ())

generatesan event stream whoseevents correspond to the moments when the
robot gets \stuck:" that is, an event is generatedevery time the robot's motion
is blocked by an obstaclethat it has run into.

What makes event streams special is that there is a special set of func-
tions that useevent streamsto achieve various kinds of switching. The simplest
switching combinator is called switch , whosetype is given by:

switch :: SF a (b, Event c) -> (c -> SF a b) -> SF a b

The expression(sf1 &&&es) `switch` \e -> sf2 behaves as sf1 until the
�rst event in the event stream es occurs, at which point the event's value is
bound to e and the behavior switchesover to sf2 .

For example, in order to prevent damageto a robot wheel's motor, we may
wish to set its speedto zero when the robot gets stuck:

xspd :: Speed -> SF SimbotInput Speed
xspd v = (constant v &&&rsStuck) `switch` \() -> constant 0

It shouldbeclear that stateful Yampaprogramscanbeconstructedusingswitch-
ing combinators.

Exercise 4. Rather than set the wheel speedto zero when the robot gets stuck,
negateit instead. Then de�ne xspd recursively so that the velocity getsnegated
every time the robot gets stuck.

6 Certain input events such as key pressesare in fact properly bu�erred in our imple-
mentation such that none will be lost.



Switc hing semantics. There are several kinds of switching combinators in
Yampa, four of which we will use in this paper. These four switchers arise out
of two choicesin the semantics:

1. Whether or not the switch happens exactly at the time of the event, or
in�nitesimally just after. In the latter case,a \ d" (for \delayed") is pre�xed
to the name switch .

2. Whether or not the switch happensjust for the �rst event in an event stream,
or for every event. In the latter case,an \ r " (for \recurring") is pre�xed to
the name switch .

This leadsto the four switchers, whosenamesand typesare:

switch, dSwitch :: SF a (b, Event c) -> (c -> SF a b) -> SF a b
rSwitch, drSwitch :: SF a b -> SF (a, Event (SF a b)) b

An example of the use of switch was given above. Delayed switching is useful
for certain kinds of recursive signal functions. In Sec.2.7 we will seean example
of the useof drSwitch .

As mentioned earlier, an important property of switching is that time begins
afreshwithin each signal function being switched into. For example,considerthe
expression:

let sinSF = time >>> arr sin
in (sinSF &&&rsStuck) `switch` const sinSF

sinSF to the left of the switch generatesa sinusoidal signal. If the �rst event
generatedby rsStuck happensat time t, then the sinSF on the right will begin
at time 0, regardlessof what the time t is; i.e. the sinusoidal signal will start
over at the time of the event.

Useful event functions. Event is an instance of class Functor , and thus
fmap can be usedto changethe value carried by an event. For example,we can
increment the value of an event e :: Event Double by writing fmap (+1) e.
Sometimeswe don't care about the old value of an event when creating a new
one, so Yampa also provides:

tag :: Event a -> b -> Event b
e `tag` b = fmap (const b) e

It is often desirableto merge events; for example, to form the disjunction of
two logical events. The only problem is deciding what to do with simultaneous
events. The most general form of merge:

mergeBy :: (a -> a -> a) -> Event a -> Event a -> Event a

allows the user to decide how to handle simultaneous events by providing a
function to combine the event values. Alternativ ely, one may choose to give
preferenceto the left or right event:



lMerge :: Event a -> Event a -> Event a
rMerge :: Event a -> Event a -> Event a

If there is no possibility of simultaneous events, merge may be used, which
generatesan error if in fact two events occur together:

merge :: Event a -> Event a -> Event a

So far we have only consideredpre-existing events. Someof thesemay come
from external sources,such as a bumper switch or communications subsystem,
but it is often convenient to de�ne our own events. Yampa provides a variety of
ways to generatenew events, the most important being:

edge :: SF Bool (Event ())

The expressionboolSF >>> edge generatesan event every time the signal from
boolSF goes from False to True (i.e. the \leading edge" of the signal). For
example,if tempSF :: SF SimbotInput Tempis a signal function that indicates
temperature, then:

alarmSF :: SF SimbotInput (Event ())
alarmSF = tempSF >>> arr (>100) >>> edge

generatesan alarm event if the temperature exceeds100 degrees.
Here are a few other useful event generation functions:

never :: SF a (Event b)
now :: b -> SF a (Event b)
after :: Time -> b -> SF a (Event b)
repeatedly :: Time -> b -> SF a (Event b)

never is an event source that never generatesany event occurrences.now v
generatesexactly oneevent, whosetime of occurrenceis zero(i.e. now) and whose
value is v. The expressionafter t v generatesexactly one event, whosetime
of occurrenceis t and whosevalue is v. Similarly, repeatedly t v generatesan
event every t seconds,each with value v.

To closethis section, we point out that the discrete and continuous worlds
interact in important ways, with switching, of course, being the most funda-
mental. But Yampa also provides several other useful functions to capture this
interaction. Here are two of them:

hold :: a -> SF (Event a) a
accum :: a -> SF (Event (a -> a)) (Event a)

The signal function hold v initially generatesa signal with constant value v,
but every time an event occurs with value v' , the signal takeson (i.e. \holds")
that new value v' . The signal function accum v0 is assentially an event stream
transformer. Each input event generatesone output event. If f n is the function
corresponding to the nth input event, then the value vn of the nth output event
is just f n vn � 1, for n � 1, and with v0 = v0.

For example, the following signal function represents the number of alarms
generatedfrom alarmSF de�ned earlier:



alarmCountSF :: SF SimbotInput Int
alarmCountSF = alarmSF >>> arr (`tag` (+1)) >>> accum 0 >>> hold 0

Indeed, the accumfollowed by hold idiom is so common that it is prede�ned in
Yampa:

accumHold :: a -> SF (Event (a -> a)) a
accumHold init = accum init >>> hold init

Exercise 5. Supposev :: SF SimbotInput Velocity represents the scalar ve-
locity of a simbot. If we integrate this velocity, we get a measureof how far the
simbot has traveled. De�ne an alarm that generatesan event when either the
simbot has traveled more than d meters, or it has gotten stuck.

2.7 Recursiv e Signals

Note in Fig. 1 the presenceof the loop combinator. Its purpose is to de�ne
recursive signal functions; i.e. it is a �xp oint operator. The arrow syntax goes
one step further by allowing recursive de�nitions to be programmed directly,
which the preprocessorexpandsinto applications of the loop combinator. In this
casethe user must include the keyword rec prior to the collection of recursive
bindings.

For example, a common needwhen switching is to take a \snapshot" of the
signal being switched out of, for use in computing the value of the signal being
switched into. Supposethat there is an event source
incrVelEvs :: SF SimbotInput (Event ()) whoseevents correspond to com-
mands to increment the velocity. We can de�ne a signal function that responds
to thesecommandsas follows:

vel :: Velocity -> SF SimbotInput Velocity
vel v0 = proc inp -> do

rec e <- incrVelEvs -< inp
v <- drSwitch (constant v0) -< (inp, e `tag` constant (v+1))

returnA -< v

Note that v is recursively de�ned. This requiresthe useof the rec keyword, and
alsothe useof a delayed switch to ensurethat the recursionis well founded.Also
note that the recurring version of switch is used, becausewe want the velocity
update to happen on every event. Finally, note the use of tag to update the
value of an event.

The need for a delayed switch is perhaps best motivated by analogy to re-
cursively de�ned lists, or streams.The de�nition:

ones = 1 : ones

expressesthe usual in�nite streamof ones,and is obviously well founded,whereas
the list:

ones = ones



is obviously not well founded. The value of 1 placed at the front of the list
can be thought of as a delay in the accessof ones. That is the idea behind a
delayed switch, although semantically the delay is intended to be in�nitesimally
small, and in the implementation we avoid intro ducing a delay that could a�ect
performance.

Exercise 6. Rede�ne vel using dSwitch instead of drSwitch , and without using
the rec keyword. (Hint: de�ne vel recursively instead of de�ning v recursively.)

3 Programming the Rob ot Simulator

3.1 Rob ot Input and Output

Generally speaking, one might have dozensof di�eren t robots, somereal, some
simulated, and each with di�eren t kinds of functionalit y (t wo wheels, three
wheels,four wheels,cameras,sonars,bumper switches,actuators, speakers,
ash-
ing lights, missle launchers, and so on). These di�erences are captured in the
input and output types of the robot. For example, there is only one kind of
simulated robot, or simbot, whoseinput type is SimbotInput and whoseoutput
type is SimbotOutput .

[Note: The code described in this sectionworks with Yampa version0.9 (and
0.9.x patches), but some changes are anticipated for use with future Yampa
versions1.0 and higher. In particular, the module nameswill change.In Yampa
0.9 they are still known under their old names(AFrob, AFrobRobotSim, etc.) for
backwards compatibilit y reasons.]

We refer to the collection of Yampa libraries that are robot-speci�c asAFrob.
The AFrob library was written to be as genericas possible,and thus it doesnot
depend directly on the robot input and output types. Rather, type classesare
used to capture di�eren t kinds of functionalit y. Each robot type is an instance
of somesubsetof theseclasses,depending on the functionalit y it has to o�er.

For example, SimbotInput is a member of the type classesshown in the
upper half of Fig. 3, and SimbotOutput is a member of the lower ones.The types
Velocity , Distance , Angle, RotVel , RotAcc, Length, Acceleration , Speed,
Heading, and Bearing are all synonyms for type Double. Type Position2 is a
synonym for Point2 Position , where:

data RealFloat a => Point2 a = Point2 !a !a
deriving Eq

We will give examplesof the useof many of theseoperations and type classes
in the examplesthat follow. Before doing so, however, there is one other detail
to describe about the output classes.Note in Fig. 3 that the methods in the last
two classesreturn a type MRa, wherea is constrainedto be a MergeableRecord.
This allows one to incrementally specify certain \�elds" of the record, and to
mergethem later. There are two key operations on mergeablerecords:

mrMerge :: MergeableRecord a => MRa -> MRa -> MRa
mrFinalize :: MergeableRecord a => MRa -> a



-- Input Classes And Related Functions
-------------------- --- -- -- -- -- --- -- -- -- -- --- -- -- -- -- --- -- -- -- -- --- -
class HasRobotStatus i where

rsBattStat :: i -> BatteryStatus -- Curent battery status
rsIsStuck :: i -> Bool -- Currently stuck or not

data BatteryStatus = BSHigh | BSLow| BSCritical
deriving (Eq, Show)

-- derived event sources:
rsBattStatChanged :: HasRobotStatus i => SF i (Event BatteryStatus)
rsBattStatLow :: HasRobotStatus i => SF i (Event ())
rsBattStatCritical :: HasRobotStatus i => SF i (Event ())
rsStuck :: HasRobotStatus i => SF i (Event ())

class HasOdometry i where
odometryPosition :: i -> Position2 -- Current position
odometryHeading :: i -> Heading -- Current heading

class HasRangeFinder i where
rfRange :: i -> Angle -> Distance
rfMaxRange :: i -> Distance

-- derived range finders:
rfFront :: HasRangeFinder i => i -> Distance
rfBack :: HasRangeFinder i => i -> Distance
rfLeft :: HasRangeFinder i => i -> Distance
rfRight :: HasRangeFinder i => i -> Distance

class HasAnimateObjectTracker i where
aotOtherRobots :: i -> [(RobotType, Angle, Distance)]
aotBalls :: i -> [(Angle, Distance)]

class HasTextualConsoleInpu t i where
tciKey :: i -> Maybe Char

tciNewKeyDown:: HasTextualConsoleInpu t i =>
Maybe Char -> SF i (Event Char)

tciKeyDown :: HasTextualConsoleInpu t i => SF i (Event Char)

-- Output Classes And Related Functions
-------------------- --- -- -- -- -- --- -- -- -- -- --- -- -- -- -- --- -- -- -- -- --- -
class MergeableRecord o => HasDiffDrive o where

ddBrake :: MRo -- Brake both wheels
ddVelDiff :: Velocity -> Velocity -> MRo -- set wheel velocities
ddVelTR :: Velocity -> RotVel -> MRo -- set vel. and rot.

class MergeableRecord o => HasTextConsoleOutput o where
tcoPrintMessage :: Event String -> MRo

Fig. 3. Robot Input and Output Classes



For example, the expression:

sbo :: SimbotOutput
sbo = mrFinalize

(ddVelDiff vel1 vel2 `mrMerge` tcoPrintMessage stringEvent)

mergesthe velocity output with a consolemessage.
For simbots, it turns out that velocity control and messageoutput are the

only two things that can be merged, so the use of the MergeableRecord class
may seemlike an overkill. However, for other robots there may be many such
mergeableoutputs, and the functionalit y thus o�ered is quite convenient.

When two common outputs are merged, the result depends on how the
mrMergeand mrFinalize methods are de�ned to behave. The designerof a par-
ticular instanceof thesemethods might signal an error, acceptoneoutput or the
other (for example,merging two calls to ddVelDiff yields the value of the �rst
one), or combine the two (for example, merging two calls to tcoPrintMessage
results in both messagesbeing printed in order).

3.2 Rob ot Con trollers

To control a robot we must de�ne a robot controller, which, for the case of
simbots, must have type:

type SimbotController =
SimbotProperties -> SF SimbotInput SimbotOutput

SimbotProperties is a data type that speci�es static properties of a simbot.
These properties are accessedabstractly in that SimbotProperties is an in-
stanceof the HasRobotProperties type class:

class HasRobotProperties i where
rpType :: i -> RobotType -- Type of robot
rpId :: i -> RobotId -- Identity of robot
rpDiameter :: i -> Length -- Distance between wheels
rpAccMax :: i -> Acceleration -- Max translational acc
rpWSMax :: i -> Speed -- Max wheel speed

type RobotType = String
type RobotId = Int

The simulator knows about two versionsof the simbot, for which each of these
properties is slightly di�eren t. The RobotType �eld is just a string, which for
the simbots will be either "SimbotA" or "SimbotB" . The remaining �elds are
self-explanatory.

To actually run the simulator, we usethe function:

runSim :: Maybe WorldTemplate ->
SimbotController -> SimbotController -> IO ()



wherea WorldTemplate is a data type that describesthe initial state of the sim-
ulator world. It is a list of simbots, walls, balls, and blocks, along with locations
of the centers of each:

type WorldTemplate = [ObjectTemplate]

data ObjectTemplate =
OTBlock { otPos :: Position2 } -- Square obstacle

| OTVWall { otPos :: Position2 } -- Vertical wall
| OTHWall { otPos :: Position2 } -- Horizontal wall
| OTBall { otPos :: Position2 } -- Ball
| OTSimbotA{ otRId :: RobotId, -- Simbot A robot

otPos :: Position2,
otHdng :: Heading }

| OTSimbotB{ otRId :: RobotId, -- Simbot B robot
otPos :: Position2,
otHdng :: Heading }

The constants worldXMin, worldYMin, worldXMax, and worldYMaxarethe bounds
of the simulated world, and are assumedto be in meters.Currently thesevalues
are -5, -5, 5, and 5, respectively (i.e. the world is 10 meters by 10 meters, with
the center coordinate being (0; 0)). The walls are currently �xed in sizeat 1:0m
by 0:1m, and the blocks are 0:5m by 0:5m. The diameter of a simbot is 0:5m.

Your overall program should be structured as follows:

module MyRobotShowwhere

import AFrob
import AFrobRobotSim

main :: IO ()
main = runSim (Just world) rcA rcB

world :: WorldTemplate
world = ...

rcA :: SimbotController -- controller for simbot A's
rcA = ...

rcB :: SimbotController -- controller for simbot B's
rcB = ...



The module AFrob also imports the Yampa library . The module AFrobRobotSim
is the robot simulator.

Note that many robots may be created of the samekind (i.e. simbot A or
simbot B) in the world template, but the same controller will be invoked for all
of them. If you want to distinguish amongst them, simply give them di�eren t
RobotID's. For example, if you have three simbot A robots, then your code for
controller rcA can be structured like this:

rcA :: SimbotController
rcA rProps =

case rpId rProps of
1 -> rcA1 rProps
2 -> rcA2 rProps
3 -> rcA3 rProps

rcA1, rcA2, rcA3 :: SimbotController
rcA1 = ...
rcA2 = ...
rcA3 = ...

3.3 Basic Rob ot Mo vement

In this section we will write a seriesof robot controllers, each of type
SimbotController . Designing controllers for real robots is both an art and a
science.The sciencepart includes the use of control theory and related math-
ematical techniques that focus on di�eren tial equations to design optimal con-
trollers for speci�c tasks.We will not spend any time on control theory here,and
instead will appeal to the reader's intuition in the design of functional, if not
optimal, controllers for mostly simple tasks. For more details on the kinematics
of mobile robots, see[3].

Stop, go, and turn. For starters, let's de�ne the world's dumbest controller {
one for a stationary simbot:

rcStop :: SimbotController
rcStop _ = constant (mrFinalize ddBrake)

Or we could make the simbot move blindly forward at a constant velocity:

rcBlind1 _ = constant (mrFinalize $ ddVelDiff 10 10)

We can do one better than this, however, by �rst determining the maximal
allowable wheelspeedsand then running the simbot at, say, one-half that speed:

rcBlind2 rps =
let max = rpWSMaxrps
in constant (mrFinalize $ ddVelDiff (max/2) (max/2))



We can also control the simbot through ddVelTR, which allows specifying
the simbot's forward and rotational velocities, rather than the individual wheel
speeds.For a di�eren tial drive robot, the maximal rotational velocity dependson
the vehicle'sforward velocity; it can rotate most quickly when it is standing still,
and cannot rotate at all if it is going at its maximal forward velocity (becauseto
turn while going at its maximal velocity, one of the wheelswould have to slow
down, in which caseit would no longer be going at its maximal velocity). If the
maximal wheel velocity is vmax , and the forward velocity is vf , then it is easy
to show that the maximal rotational velocity in radians per secondis given by:

! max =
2(vmax � vf )

l

For example, this simbot turns as fast as possiblewhile going at a given speed:

rcTurn :: Velocity -> SimbotController
rcTurn vel rps =

let vMax = rpWSMaxrps
rMax = 2 * (vMax - vel) / rpDiameter rps

in constant (mrFinalize $ ddVelTR vel rMax)

Exercise 7. Link rcBlind2 , rcTurn , and rcStop together in the following way:
Perform rcBlind2 for 2 seconds,then rcTurn for three seconds,and then do
rcStop . (Hint: useafter to generatean event after a given time interval.)

The sim bot talks (sort of ). For something more interesting, let's de�ne
a simbot that, whenever it gets stuck, reversesits direction and displays the
message"Ouch!!" on the console:

rcReverse :: Velocity -> SimbotController
rcReverse v rps = beh `dSwitch` const (rcReverse (-v) rps)

where beh = proc sbi -> do
stuckE <- rsStuck -< sbi
let mr = ddVelDiff v v `mrMerge`

tcoPrintMessage (tag stuckE "Ouch!!")
returnA -< (mrFinalize mr, stuckE)

Note the useof a let binding within a proc : this is analogousto a let binding
within Haskell's monadic do syntax. Note also that rcReverse is recursive {
this is how the velocity is reversedeverytime the simbot gets stuck { and there-
fore requires the use of dSwitch to ensure that the recursion is well founded.
(It does not require the rec keyword, however, becausethe recursion occurs
outside of the proc expression.)The other reasonfor the dSwitch is rather sub-
tle: tcoPrintMessage usesstuckE to control when the messageis printed, but
stuckE also controls the switch; thus if the switch happened instantaneously,
the messagewould be missed!



If preferred, it is not hard to write rcReverse without the arrow syntax:

rcReverse' v rps =
(rsStuck >>> arr fun) `dSwitch` const (rcReverse' (-v) rps)
where fun stuckE =

let mr = ddVelDiff v v `mrMerge`
tcoPrintMessage (tag stuckE "Ouch!!")

in (mrFinalize mr, stuckE)

Exercise 8. Write a version of rcReverse that, instead of knowing in advance
what its velocity is, takesa \snapshot" of the velocity, as described in Sec.2.7,
at the moment the stuck event happens,and then negatesthis value to continue.

Finding our way using odometry . Note from Fig. 3 that our simbots have
odometry; that is, the abilit y of a robot to track its own location. This capability
on a real robot can be approximated by so-called\dead reckoning," in which the
robot monitors its actual wheel velocities and keepstrack of its position incre-
mentally . Unfortunately , this is not particularly accurate, becauseof the errors
that arise from wheel slippage,uneven terrain, and so on. A better technique is
to useGPS (global positioning system), which usessatellite signalsto determine
a vehicle's position to within a few feet of accuracy. In our simulator we will
assumethat the simbot's odometry is perfect.

We can useodometry readingsas feedback into a controller to stabilize and
increasethe accuracy of somedesired action. For example, supposewe wish to
move the simbot at a �xed speed in a certain direction. We can set the speed
easily enough as shown in the examplesabove, but we cannot directly specify
the direction. However, we can read the direction using the odometry function
odometryHeading :: SimbotInput -> Heading and usethis to control the ro-
tational velocity.

(A note about robot headings.In AFrob there are three data typesthat relate
to headings:

1. Heading is assumedto be in radians, and is aligned with the usual Cartesian
coordinate system, with 0 radians corresponding to the positive x-axis, � =2
the positive y-axis, and so on. Its normalized range is [� � ; � ).

2. Bearing is assumedto be in degrees,and is aligned with a conventional
compass,with 0 degreescorresponding to north, 90 degreesto east, and so
on. Its normalized range is [0; 360).

3. Angle is assumedto be in radians, but is a relative measurerather than
being aligned with something absolute.

AFrob also provide conversion functions betweenbearingsand headings:

bearingToHeading :: Bearing -> Heading
headingToBearing :: Heading -> Bearing

However, in this paper we only useheadingsand relative angles.)



Getting back to our problem, if hd and ha are the desiredand actual headings
in radians, respectively, then the heading error is just he = hd � ha . If he is
positive, then we want to turn the robot in a counter-clockwise direction (i.e.
using a positive rotational velocity), and if he is negative, then we want to turn
the robot in a clockwise direction (i.e. using a negative rotational velocity). In
other words, the rotational velocity should be directly proportio onal to he (this
strategy is thus called a proportionate controller ). One small complication to
this schemeis that we needto normalize hd � ha to keepthe angle in the range
[� � ; � ). This is easily achieved using Yampa's normalizeAngle function. Here
is the complete controller:

rcHeading :: Velocity -> Heading -> SimbotController
rcHeading vel hd rps =

let vMax = rpWSMaxrps
vel' = lim vMax vel
k = 2

in proc sbi -> do
let he = normalizeAngle (hd - odometryHeading sbi)
let vel'' = (1 - abs he / pi) * vel'
returnA -< mrFinalize (ddVelTR vel'' (k*he))

lim m y = max (-m) (min m y)

The parameter k is called the gain of the controller, and can be adjusted to give
a faster response,at the risk of being too fast and thus being unstable. lim m y
limits the maximum absolute value of y to m.

Before the next example we will �rst rewrite the above program in the fol-
lowing way:

rcHeading' :: Velocity -> Heading -> SimbotController
rcHeading' vel hd rps =

proc sbi -> do
rcHeadingAux rps -< (sbi, vel, hd)

rcHeadingAux :: SimbotProperties ->
SF (SimbotInput,Velocity,Heading) SimbotOutput

rcHeadingAux rps =
let vMax = rpWSMaxrps

k = 2
in proc (sbi,vel,hd) -> do

let vel' = lim vMax vel
let he = normalizeAngle (hd - odometryHeading sbi)
let vel'' = (1 - abs he / pi) * vel'
returnA -< mrFinalize (ddVelTR vel'' (k*he))

In the original de�nition, vel and hd were constant during the lifetime of the
signal function, whereas in the second version they are treated as signals in



rcHeadingAux, thus allowing for them to be time varying. Although not needed
in this example,we will needthis capability below.

As another exampleof using odometry, considerthe task of moving the sim-
bot to a speci�c location. We can do this by computing a tra jectory from our
current location to the desired location. By doing this continually, we ensure
that drift causedby imperfections in the robot, the 
o or surface, etc. do not
causeappreciableerror.

The only complication is that we must take into account our simbot's trans-
lational inertia: if we don't, we may overshoot the target. What we'd like to
do is slow down as we approach the target (as for rcHeading , this amounts to
designinga proportionate controller). Here is the code:

rcMoveTo :: Velocity -> Position2 -> SimbotController
rcMoveTo vd pd rps = proc sbi -> do

let (d,h) = vector2RhoTheta (pd .-. odometryPosition sbi)
vel = if d>2 then vd else vd*(d/2)

rcHeadingAux rps -< (sbi, vel, h)

Note the use of vector arithmetic to compute the di�erence between the de-
sired position pd and actual position odometryPosition sbi , and the use of
vector2RhoTheta to convert the error vector into distance d and heading h.
vel is the speed at which we will approach the target. Finally, note the use
of rcHeadingAux de�ned above to move the simbot at the desiredvelocity and
heading.

Exercise 9. rcMoveTowill behave a little bit funny once the simbot reaches its
destination, becausea di�eren tial drive robot is not able to maneuver well at
slow velocities (compare the di�cult y of parallel parking a car to the easeof
switching lanes at high speed). Modify rcMove so that once it gets reasonably
closeto its target, it stops (using rcStop ).

Exercise 10. De�ne a controller to causea robot to follow a sinusoidal path.
(Hint: feeda sinusoidal signal into rcHeadingAux.)

Exercise 11. De�ne a controller that takesa list of points and causesthe robot
to move to each point successively in turn.

Exercise 12. (a) De�ne a controller that chasesa ball. (Hint: use the aotBalls
method in classHasAnimateObjectTracker to �nd the location of the ball.) (b)
Once the ball is hit, the simulator will stop the robot and create an rsStuck
event. Therefore, modify your controller so that it restarts the robot whenever
it gets stuck, or perhapsbacks up �rst and then restarts.

Home on the range. Recall that our simbots have range�nders that are able
to determine the distance of the nearest object in a given direction. We will
assumethat there are four of these,one looking forward, one backward, one to
the left, and one to the right:



rfFront :: HasRangeFinder i => i -> Distance
rfBack :: HasRangeFinder i => i -> Distance
rfLeft :: HasRangeFinder i => i -> Distance
rfRight :: HasRangeFinder i => i -> Distance

Theseare intended to simulate four sonarsensors,except that they are far more
accurate than a conventional sonar, which has a rather broad signal. They are
more similar to the capability of a laser-basedrange �nder.

With a range �nder we can do some degreeof autonomous navigation in
\unkno wn terrain." That is, navigation in an area where we do not have a
precisemap. In such situations a certain degreeof the navigation must be done
based on local features that the robot \sees," such as walls, doors, and other
objects.

For example, let's de�ne a controller that causesour simbot to follow a wall
that is on its left. The idea is to move forward at a constant velocity v, and as
the desireddistance d from the wall varies from the left range �nder reading r ,
adjustments are made to the rotational velocity ! to keep the simbot in line.
This task is not quite as simple as the previous ones,and for reasonsthat are
beyond the scope of this paper, it is desirable to use what is known as a PD
(for \prop ortionate/deriv ative") controller, which meansthat the error signal is
fed back proportionately and alsoas its derivative. More precisely, onecan show
that, for small deviations from the norm:

! = K p(r � d) + K d(
dr
dt

)

K p and K d arethe proportionate gain and derivativegain, respectively. Generally
speaking, the higher the gain, the better the reponsewill be, but care must be
taken to avoid responding too quickly, which may causeover-shooting the mark,
or worse,unstable behavior that is oscillatory or that diverges.It can be shown
that the optimal relationship betweenK p and K d is given by:

K p = vK 2
d =4

In the code below, we will set K d to 5. For pragmatic reasonswe will also put a
limit on the absolute value of ! using the limiting function lim .

Assuming all of this mathematics is correct, then writing the controller is
fairly straightforward:

rcFollowLeftWall :: Velocity -> Distance -> SimbotController
rcFollowLeftWall v d _ = proc sbi -> do

let r = rfLeft sbi
dr <- derivative -< r
let omega= kp*(r-d) + kd*dr

kd = 5
kp = v*(kd^2)/4

returnA -< mrFinalize (ddVelTR v (lim 0.2 omega))



Exercise 13. Enhance the wall-follower controller so that it can make left and
right turns in a mazeconstructed only of horizontal and vertical walls. Speci�-
cally:

1. If the simbot seesa wall directly in front of itself, it should slow down as it
approaches the wall, stopping at distance d from the wall. Then it should
turn right and continue following the wall which should now be on its left.
(This is an inside-cornerright turn.)

2. If the simbot losestrack of the wall on its left, it continuesstraight aheadfor
a distance d, turns left, goesstraight for distance d again, and then follows
the wall which should again be on its left. (This is an outside-corner left
turn.)

Test your controller in an appropriately designedworld template.

Exercise 14. As mentioned in the derivation above, the rcFollowLeftWall con-
troller is only useful oncethe robot is closeto being on track: i.e. at the proper
distance from the wall and at the proper heading. If the robot is too far from
the wall, it will tend to turn too much in trying to get closer,which makes the
left range �nder seean even greater distance,and the systembecomesunstable.
Designing a more robust wall follower is tric ky business,and is best treated as
multi-mode system,wherethe robot �rst seeksa wall, aligns itself parallel to the
wall, and then tries to follow it. Design such a controller.

Mass hysteria. As mentioned earlier, the simulator can handle a number of
simbots simultaneously. Groups of robots can exhibit all kinds of interesting and
productive group behavior (or possibly masshysteria), limited only by the clev-
ernessof you, the designer.We will describe one simple kind of group behavior
here, leaving others (such as the soccer match described in Ex. 16) to you.

The behavior that we will de�ne is that of convergence. Assumethat all sim-
bots are initially moving in arbitrary directions and speeds.Each simbot will
look at the positions of all of the others, and move toward the centroid (i.e. av-
erage)of thosepositions. If each robot doesthis continuously and independently ,
they will all end up converging upon the samepoint.

To achieve this, recall �rst the HasAnimateObjectTracker class:

class HasAnimateObjectTracker i where
aotOtherRobots :: i -> [(RobotType, RobotId, Angle, Distance)]
aotBalls :: i -> [(Angle, Distance)]

The �rst of theseoperations permits us to determine the angle and distance of
each of the other simbots. By converting thesemeasurements to vectors, we can
add them and take their average,then usercHeading to steer the robot toward
the resulting centroid.

Other than dealing with numeric conversions,the �nal code is fairly straight-
forward:



rcAlign :: Velocity -> SimbotController
rcAlign v rps = proc sbi -> do

let neighbors = aotOtherRobots sbi
vs = map (\(_,_,a,d) -> vector2Polar d a) neighbors
avg = if vs==[] then zeroVector

else foldl1 (^+^) vs ^/ intToFloat (length vs)
heading = vector2Theta avg + odometryHeading sbi

rcHeadingAux rps -< (sbi, v, heading)
intToFloat = fromInteger . toInteger

When observing the world through robot sensors,one should not make too
many assumptionsabout what one is going to see,becausenoise, varying light
conditions, occlusion, etc. can destroy those expectations. For example, in the
caseof the simbots, the simulator doesnot guarantee that all other robots will
be visible through the animate object tracker. Indeed,at the very �rst time-step,
none are visible. For reasonsof causality, sensordata is delayed one time-step;
but at the very �rst time step, there is no previous data to report, and thus the
animate object tracker returns an empty list of other robots. This is why in the
code above the list vs is tested for being empty.

Exercise 15. Write a program for two simbots that are traveling in a straight
path, except that their paths continually interleave each other, as in a braid of
rope. (Hint: treat the velocities as vectors, and determine the proper equations
for two simbots to circle one another while maintaining a speci�ed distance.
Then add thesevelocities to the simbots' forward velocities to yield the desired
behavior.)

Exercise 16. Write a program to play \rob ocup soccer," as follows. Using wall
segments, create two goals at either end of the �eld. Decide on a number of
players on each team, and write controllers for each of them. You may wish to
write a coupleof genericcontrollers, such asonefor a goalkeeper, onefor attack,
and one for defense.Create an initial world where the ball is at the center mark,
and each of the players is positioned strategically while being on-side(with the
defensive players also outside of the center circle). Each team may usethe same
controller, or di�eren t ones. Indeed, you can pit your controller-writing skills
against those of your friends (but we do not recommendbetting money on the
game'soutcome).

